

Tema 9

Càlcul integral de funcions reals de variable real

Objectius:

1. Calcular funcions primitives amb wxMaxima.
2. Practicar amb el concepte de funció integrable i la integral d'una funció.
3. Treballar amb funcions definides mitjançant integrals
4. Calcular integrals de forma aproximada.
5. Estudiar algunes aplicacions de la integral.
6. Estudiar la generalització de la integral (integrals impròpies).

Continguts:

- 09-1. Antiderivades o primitives d'una funció. Integral indefinida.
- 09-2. Integral d'una funció en un interval. Càlcul d'integrals.
- 09-3. La funció integral. Aplicacions
- 09-4. Càlcul aproximat d'integrals. Integració numèrica.
- 09-5. Integrals impròpies.

Referències

- AEM11 ALANINOS PRATS, J; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P. (2011)
Cálculo con wxMaxima.
- APJ11 ALANINOS PRATS, J; APARICIO DEL PRADO, C; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P.; VILLENA MUÑOZ, A.R. (2011)
Prácticas de ordenador con wxMaxima.
- AP86 APOSTOL, T.M. (1986)
Análisis Matemático

- BR09 BRUZÓN GALLEGO, M. DE LOS SANTOS; RAMÍREZ LABRADOR, JOSÉ (2009)
Modelos matemáticos con Maxima
- BU07 DE BURGOS, JUAN (2007)
Cálculo Infinitesimal de una variable (segunda edición).
- ES08 ESTELA CARBONELL, M. ROSA; SAÀ SEOANE, JOEL (2008)
Cálculo con soporte interactivo en moodle.
- ES02 ESTEP, DONALD (2002)
Practical Analysis in one variable
- GV07a GONZÁLEZ VEGA, LAUREANO (2007)
Laboratorio de Matemáticas. Vol. 1: números y ecuaciones
- GV07b GONZÁLEZ VEGA, LAUREANO (2007)
Laboratorio de Matemáticas. Vol. 2: límites y derivadas
- JB01 JARAUTA BRAGULAT, EUSEBI (2001)
Anàlisi Matemàtica d'una variable. Fonaments i aplicacions.
- RR05 REDONDO NEBLE, M. VICTORIA; RODRÍGUEZ GALVÁN, J. RAFAEL (2005)
Introducción a Maxima
- RR08b RODRÍGUEZ RIOTORTO, MARIO (2008)
Curso intensivo i-MATH de software libre orientado a Ciencias e Ingeniería
- RU80 RUDIN, WALTER (1980)
Principios de Análisis Matemático.
- SP95 SPIVAK, MICHAEL (1995)
Calculus (Càlcul Infinitesimal).
- VR09 VALLEJO RODRÍGUEZ, JOSÉ ANTONIO (2009)
Cálculo diferencial con Maxima

09-1.- Antiderivades o primitives d'una funció. Integral indefinida.

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_09-1.wxm**.

Definició. Si $f:[a,b] \subset \mathbb{R} \rightarrow \mathbb{R}$ és una funció acotada en un interval $[a,b]$, s'anomena antiderivada o primitiva de f a qualsevol funció F contínua en $[a,b]$, derivable en $]a,b[$ i tal que $DF(x)=f(x)$ per a tot $x \in]a,b[$. Així per exemple, $F_1(x)=x^3+2$ i $F_2(x)=x^3-6$ són antiderivades de la funció $f(x)=3x^2$ en qualsevol interval $[a,b] \subset \mathbb{R}$.

Si F és una antiderivada de f , és immediat raonar que el conjunt de totes les antiderivades de f s'anomena integral indefinida de f i es defineix mitjançant:

$$D^{-1}(f(x)) = \{F(x) + C, DF(x) = f(x), C = \text{constant}\}$$

També és habitual designar la integral indefinida d'una funció amb la notació clàssica $\int f(x)dx$, notació que es justificarà més endavant.

La metodologia clàssica per a l'obtenció de la integral indefinida d'una funció es fonamenta en dos resultats importants: la fórmula d'integració per parts i el teorema del canvi de variable, que enunciem a continuació.

Càlcul de primitives per parts. Si es tracta de calcular la integral indefinida d'una funció $f(x)$, es pot escriure aquesta funció com el producte $f(x)=u(x)Dv(x)$ i aleshores es pot calcular la integral indefinida de f mitjançant l'expressió:

$$D^{-1}(f(x)) = D^{-1}(u(x)Dv(x)) = u(x)v(x) - D^{-1}(Du(x)v(x)) + C$$

La notació clàssica d'aquest mètode és:

$$\int f(x)dx = \int u(x)Dv(x)dx = u(x)v(x) - \int Du(x)v(x)dx + C$$

Càlcul de primitives per canvi de variable. Per calcular la integral indefinida d'una funció contínua $f:[a,b] \subset \mathbb{R} \rightarrow \mathbb{R}$ es considera una funció $\varphi:[c,d] \rightarrow [a,b]$ bijectiva i derivable en l'interval $]c,d[$ tal que $D\varphi(t) \neq 0$ per a tot $t \in]c,d[$. Aquesta funció s'anomena funció de canvi de variable. Per tant, si $x = \varphi(t)$ això equival a $t = \varphi^{-1}(x)$ i a més la funció inversa φ^{-1} és derivable en $]a,b[$, verificant-se que

$$D\varphi^{-1}(x) = \frac{1}{D\varphi(t)}$$

Si es considera ara una antiderivada G de la funció $g(t) = f(\varphi(t))D\varphi(t)$ es compleix que:

$$D(G \circ \varphi^{-1})(x) = DG(\varphi^{-1}(x)) \cdot D\varphi^{-1}(x) = f(\varphi(t))D\varphi(t) \frac{1}{D\varphi(t)} = f(x)$$

Així doncs, el càlcul d'una antiderivada de f es pot dur a terme calculant una antiderivada de la funció $g(t)$ i aleshores es compleix:

$$D^{-1}(f(x)) = (D^{-1}(g(t))) \circ \varphi^{-1}(x) + C$$

expressió que es coneix com antiderivació o primitivació per canvi de variable o per substitució.

La notació clàssica d'aquest mètode és:

$$\int f(x)dx = \left(\int f(g(t))Dg(t)dt \right) \circ \varphi^{-1}(x) + C$$

Anem a veure com es desenvolupa el càlcul d'antiderivades o primitives amb wxMaxima. En primer lloc cal definir la funció a integrar i després la instrucció que cal aplicar és “integrate” especificant la funció i la variable. Així, per exemple:

```
(%i1) f1(x):=2*x^2-5*x+1; integrate(f1(x), x);
(%o1) f1(x):= 2 x^2 - 5 x + 1
(%o2)  $\frac{2 x^3}{3} - \frac{5 x^2}{2} + x$ 
```

El resultat és la primitiva que podríem anomenar “canònica”, ja que la constant d'integració és nul·la. Si es vol una notació en la qual es vegi aquest procés de forma més clara, aleshores:

```
(%i3) 'integrate(f1(x), x)=integrate(f1(x), x);
(%o3)  $\int 2 x^2 - 5 x + 1 dx = \frac{2 x^3}{3} - \frac{5 x^2}{2} + x$ 
```

També es pot calcular la integral indefinida de funcions en l'expressió de les quals hi ha paràmetres. Per exemple:

```
(%i4) f2(x):=a[2]*x^2+a[1]*x+a[0];
      'integrate(f2(x), x)=integrate(f2(x), x);
(%o4) f2(x):=a2 x2+a1 x+a0
(%o5)  $\int a_2 x^2 + a_1 x + a_0 dx = \frac{a_2 x^3}{3} + \frac{a_1 x^2}{2} + a_0 x$ 
```

Vegem un altre exemple:

```
(%i6) assume(a>0)$
      f3(x):=x^a;      'integrate(f3(x), x)=integrate(f3(x), x);
(%o7) f3(x):=xa
(%o8)  $\int x^a dx = \frac{x^{a+1}}{a+1}$ 
(%i9) f33(x):=x^(-1);      'integrate(f33(x), x)=integrate(f33(x), x);
(%o9) f33(x):=x-1
(%o10)  $\int \frac{1}{x} dx = \log(x)$ 
```

En ocasions el programa no dóna la resposta que es podria esperar; així, per exemple:

```
(%i11) f4(x):=log(1+sqrt(1+x^2))$
      'integrate(f4(x), x)=integrate(f4(x), x);
(%o12)  $\int \log(\sqrt{x^2+1}+1) dx = \int \frac{x^2}{(x^2+1)^{3/2}+x^2+1} dx + x \log(\sqrt{x^2+1}+1) + \operatorname{atan}(x) - x$ 
```

Cal doncs avaluar d'alguna manera la integral indefinida que apareix al segon membre. Vegem el seu càlcul; en primer lloc definim la funció a integrar:

```
(%i13) R(x):=x^2/((x^2+1)^(3/2)+x^2+1)$      integrate(R(x), x);
(%o14)  $\int \frac{x^2}{(x^2+1)^{3/2}+x^2+1} dx$ 
```

Ara calcularem una expressió d'aquesta funció aplicant una simplificació algebraica d'aquesta funció racional:

```
(%i15) r1:=x^2/((x^2+1)^(3/2)+x^2+1)$      r1=ratsimp(r1),algebraic;
(%o16)  $\frac{x^2}{(x^2+1)^{3/2}+x^2+1} = \frac{\sqrt{x^2+1}-1}{x^2+1}$ 
```

I ara ja podem expressar-la com a funció i calcular la seva integral indefinida:

```
(%i17) r2(x):=(sqrt(x^2+1)-1)/(x^2+1)$
'integrate(r2(x), x)=integrate(r2(x), x);
```

$$(\%o18) \int \frac{\sqrt{x^2+1}-1}{x^2+1} dx = \operatorname{asinh}(x) - \operatorname{atan}(x)$$

Finalment ja podem expressar la integral indefinida que es volia calcular:

```
(%i19) 'integrate(f4(x), x)=asinh(x)-atan(x)+x*log(sqrt(x^2+1)+1)+atan(x)-x;
```

$$(\%o19) \int \log(\sqrt{x^2+1}+1) dx = x \log(\sqrt{x^2+1}+1) + \operatorname{asinh}(x) - x$$

Verifiquem la correcció dels càlculs, veient que en efecte, la derivada de la funció primitiva és la funció inicial:

```
(%i20) F4(x):=x*log(sqrt(x^2+1)+1)+asinh(x)-x$
diff(F4(x), x);
```

$$(\%o21) \log(\sqrt{x^2+1}+1) + \frac{x^2}{\sqrt{x^2+1}(\sqrt{x^2+1}+1)} + \frac{1}{\sqrt{x^2+1}} - 1$$

```
(%i22) radcan(%);
```

$$(\%o22) \log(\sqrt{x^2+1}+1)$$

Vegem alguns exemples més:

```
(%i23) f5(x):=1/((x-2)*sqrt(x+2))$
'integrate(f5(x), x)=integrate(f5(x), x);
```

$$(\%o24) \int \frac{1}{(x-2)\sqrt{x+2}} dx = 2 \left(\frac{\log(\sqrt{x+2}-2)}{4} - \frac{\log(\sqrt{x+2}+2)}{4} \right)$$

```
(%i25) f6(x):=(2*x+a)/(x^2-5*x+3)$
'integrate(f6(x), x)=integrate(f6(x), x);
```

$$(\%o26) \int \frac{2x+a}{x^2-5x+3} dx = \frac{(a+5) \log\left(\frac{2x-\sqrt{13}-5}{2x+\sqrt{13}-5}\right)}{\sqrt{13}} + \log(x^2-5x+3)$$

```
(%i27) f7(x):=(b*x+a)/(x^2-5*x+3)$
'integrate(f7(x), x)=integrate(f7(x), x);
```

$$(\%o28) \int \frac{bx+a}{x^2-5x+3} dx = \frac{(5b+2a) \log\left(\frac{2x-\sqrt{13}-5}{2x+\sqrt{13}-5}\right)}{2\sqrt{13}} + \frac{b \log(x^2-5x+3)}{2}$$

```
(%i29) ratsimp(%);
```

$$(\%029) \int \frac{b x + a}{x^2 - 5 x + 3} dx = \frac{(5 b + 2 a) \log\left(\frac{2 x - \sqrt{13} - 5}{2 x + \sqrt{13} - 5}\right) + \sqrt{13} b \log(x^2 - 5 x + 3)}{2 \sqrt{13}}$$

En alguns casos el càlcul de la integral indefinida no es pot dur a terme, ja que malgrat que existeixi no es pot expressar en termes de funcions elementals. Vegem un exemple clàssic al respecte:

```
(%i30) f8(x):=exp(-x^2)$
'integrate(f8(x), x)=integrate(f8(x), x);
```

$$(\%031) \int e^{-x^2} dx = \frac{\sqrt{\pi} \operatorname{erf}(x)}{2}$$

09-2.- Integral d'una funció en un interval. Càlcul d'integrals.

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_09-2.wxm**.

Si $f : I = [a, b] \subset \mathbb{R} \rightarrow \mathbb{R}$ és una funció positiva i contínua i, per tant, acotada en l'interval tancat $[a, b]$, la integral definida, o senzillament la integral, de la funció en l'interval es calcula mitjançant el Teorema fonamental del Càlcul o regla de Barrow, el qual estableix que:

$$\int_{[a,b]} f(x)dx \equiv \int_a^b f(x)dx = F(b) - F(a) = [F(x)]_a^b$$

on F és una antiderivada o primitiva de f en $[a, b]$.

Si $f : I = [a, b] \subset \mathbb{R} \rightarrow \mathbb{R}$ és una funció positiva i contínua a trossos en l'interval tancat $[a, b]$, amb discontinuïtats evitables o de salt en aquest interval, és a dir, si $I_k = [x_k, x_{k+1}]$, $k = 1, 2, \dots, p$, $I_1 \cup \dots \cup I_p = I$, i els límits laterals de la funció en els punts frontera d'aquests intervals són finits, si indiquem per $f(x) = f_k(x)$, $x \in I_k$, $k = 1, 2, \dots, p$, i si $D^{-1}(f_k(x)) = F_k(x)$, $x \in I_k$, aleshores:

$$\int_{[a,b]} f(x)dx = \sum_{k=1}^p \int_{[x_k, x_{k+1}]} f_k(x)dx$$

$$\int_{[x_k, x_{k+1}]} f_k(x)dx = F_k(x_{k+1}) - F_k(x_k), \quad k = 1, 2, \dots, p$$

La sintaxi per al càlcul de la integral d'una funció en un interval tancat amb wxMaxima és molt senzilla i consisteix en definir la funció i aplicar la instrucció "integrate" indicant la funció, la variable i els límits inferior i superior de l'interval d'integració:

```
(%i1) f1(x):=3*x^2 - 2*x + 4$
'integrate(f1(x), x, 1, 3)=integrate(f1(x), x, 1, 3);
```

$$(\%o2) \int_1^3 3x^2 - 2x + 4 dx = 26$$

```
(%i3) f2(x):=sin(omega*x)$
'integrate(f2(x), x, a, b)=integrate(f2(x), x, a, b);
```

$$(\%o4) \int_a^b \sin(\omega x) dx = \frac{\cos(a \omega)}{\omega} - \frac{\cos(b \omega)}{\omega}$$

```
(%i5) f3(x):=exp(alpha*x)$
'integrate(f3(x), x, a, b)=integrate(f3(x), x, a, b);
```

$$(\%o6) \int_a^b e^{\alpha x} dx = \frac{e^{\alpha b}}{\alpha} - \frac{e^{\alpha a}}{\alpha}$$

Tanmateix, no sempre s'obté el resultat que s'espera:

```
(%i7) f4(x):=cos(x)/(100 - x^6)$
'integrate(f4(x), x, 0, 1)=integrate(f4(x), x, 0, 1);
```

$$(\%o8) \int_0^1 \frac{\cos(x)}{100-x^6} dx = \int_0^1 \frac{\cos(x)}{100-x^6} dx$$

És a dir, el programa no sempre calcula la integral de qualsevol funció i, per tant, cal desenvolupar altres metodologies que permetin aquest càlcul. La millor és la que s'anomena integració numèrica o càlcul aproximat d'integrals, que es veurà més endavant. Aquí podem obtenir una primera aproximació aplicant el Teorema del valor mitjà integral, el qual estableix que


$$\int_a^b f(x)dx = \mu(b-a), \quad m = \min_{[a,b]} f(x) \leq \mu \leq M = \max_{[a,b]} f(x)$$

Es defineix el valor mitjà integral mitjançant:

$$\mu = \frac{1}{(b-a)} \int_a^b f(x)dx$$

En el cas anterior ens ajudem d'una representació gràfica:

```
(%i3) wxplot2d([f(x)], [x,0, 1], [y, 0, 0.015], [ylabel, ""], [legend, "f(x)"], [style, [lines, 2]],
[gnuplot_preamble, "set size ratio 1; set grid"], [nticks, 300])$
```



S'observa que la funció és decreixent i per tant, el seu màxim s'assoleix a l'extrem inferior de l'interval i el mínim a l'extrem superior de l'interval:

```
(%i4) M=M:f(0);      m=m:f(1);
      f(0), numer;    f(1), numer;

(%o4) M =  $\frac{1}{100}$ 

      (%o5) m =  $\frac{\cos(1)}{99}$ 

      (%o6) 0.01
      (%o7) 0.0054575990491731
```

Prenent com a valor intermedi la mitjana aritmètica del màxim i del mínim, resulta:


```
(%i8) Integral_aproximada=IA:(m+M)/2;

      (%o8) Integral_aproximada =  $\frac{\frac{\cos(1)}{99} + \frac{1}{100}}{2}$ 

      (%i9) %, numer;
      (%o9) Integral_aproximada = 0.0077287995245866
```

La representació gràfica és:

```
(%i10) IA(x):=0.0077287995245866$
      wxplot2d([f(x), IA(x)], [x,0, 1], [y, 0, 0.015], [legend, "f(x)", "IA(x)"], [ylabel, ""], [style, [lines, 2]],
      [gnuplot_preamble, "set size ratio 1; set grid"], [nticks, 300])$
```



En el cas d'una funció periòdica sinusoidal, s'anomena valor eficaç de la funció l'arrel quadrada del valor mitjà integral del quadrat de la funció, és a dir:

$$V_{ef} = \sqrt{\frac{1}{(b-a)} \int_a^b (f(x))^2 dx}$$

Il·lustrem-ho gràficament amb un exemple:

```
(%i12) g(x):=(sin(x))^2$
'integrate(g(x), x, 0, 2*%pi)=integrate(g(x), x, 0, 2*%pi);
```

```
(%o13) 
$$\int_0^{2\pi} \sin(x)^2 dx = \pi$$

```


```
(%i14) Vef=Vef:sqrt((1/(2*%pi))*integrate(g(x), x, 0, 2*%pi));
```

```
(%o14) 
$$Vef = \frac{1}{\sqrt{2}}$$

```

```
(%i15) wxplot2d([g(x), Vef], [x, 0, 2*%pi], [y, 0, 1.5], [legend, "g(x)", "Vef"], [ylabel, ""], [style, [lines, 2]],
[gnuplot_preamble, "set size ratio 1; set grid"], [nticks, 300])$
```

(%t15)



09-3.- La funció integral. Aplicacions.

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_09-3.wxm**.

Si $f : I = [a, b] \subset \mathbb{R} \rightarrow \mathbb{R}$ és una funció integrable en un interval tancat i fitat, es defineix la funció integral o funció àrea mitjançant:

$$F(x) = \begin{cases} \int_a^x f(t)dt, & \text{si } a < x \leq b \\ 0, & \text{si } x = a \end{cases}$$

Aquesta funció té dues propietats molt importants:

- Es contínua en l'interval $[a, b]$.
- Si la funció f és contínua en l'interval $[a, b]$, aleshores la funció integral F és derivable en $]a, b[$ verificant-se que $DF(x) = f(x)$, $\forall x \in]a, b[$.

Amb wxMaxima es pot definir la funció integral com s'exposa a continuació. En primer lloc definir la funció, fent servir una variable diferent i a continuació calcular la integral definida posant com a límit inferior l'extrem inferior de l'interval i com a extrem superior la variable que s'assignarà a la funció integral. Vegem la sintaxi:

```
(%i1) F(x):=integrate(f(t), t, a, x);
```

```
(%o1) F(x):=  $\int_a^x f(t)dt$ 
```

Exemple 9.3.1. Considerem la funció $f : I = [-3, 3] \subset \mathbb{R} \rightarrow \mathbb{R}$ definida per:

$$f(t) = 2t^2 - 3t + 4, \quad t \in [-3, 3]$$

En primer lloc ho expressem mitjançant una integral:

```
(%i1) f(t):=2*t^2 - 3*t + 4$  
'( integrate(f(t), t, -3, x) ) = integrate(f(t), t, -3, x) ;
```

```
(%o2)  $\int_{-3}^x f(t)dt = \frac{4x^3 - 9x^2 + 24x}{6} + \frac{87}{2}$ 
```

Ara definim la funció integral:


```
(%i3) define (F(x) , integrate(f(t), t, -3, x)) ;
```

```
(%o3) F(x):=
$$\frac{4 x^3 - 9 x^2 + 24 x}{6} + \frac{87}{2}$$

```

Una representació gràfica conjunta amb les dues funcions pot ajudar a la interpretació; observi's que per fer la representació gràfica cal redefinir la funció f fent servir la mateixa variable que per a la funció integral. Així:

```
(%i3) f(x):=2*x^2 - 3*x + 4$
wxplot2d([f(x),F(x)], [x, -3, 3], [y, 0, 70] , [legend, "f(x)" , "F(x)"] , [style, [lines, 2]] ,
[plot_format, gnuplot], [gnuplot_preamble, "set size ratio 1; set grid"] , [nticks, 300] )$
```



Atès que la funció f és contínua, la funció integral F és derivable y s'ha de complir la propietat enunciada anteriorment. En efecte, calculem aquesta derivada:

```
(%i5) diff(F(x), x);
```

```
(%o5) 
$$\frac{12 x^2 - 18 x + 24}{6}$$

```

```
(%i6) ratsimp(%);
```

```
(%o6) 
$$2 x^2 - 3 x + 4$$

```

Observi's que el valor de la funció integral en l'extrem superior de l'interval ha de coincidir amb la integral de la funció en l'interval. En efecte:

```
(%i7) 'integrate(f(t), t, -3, 3)=integrate(f(t), t, -3, 3);
'F(3)=F(3);
```

```
(%o7) 
$$\int_{-3}^3 2 t^2 - 3 t + 4 dt = 60$$

```

```
(%o8) F(3)=60
```

Exemple 9.3.2. Considerem la funció $f : I = [0, 4] \subset \mathbb{R} \rightarrow \mathbb{R}$ definida per:


$$f(x) = \begin{cases} 6, & \text{si } 0 \leq x < 2 \\ 3, & \text{si } 2 \leq x \leq 4 \end{cases}$$

Aquesta funció no és contínua i observarem que la funció integral és contínua però no derivable en l'interval de definició. En efecte:

```
(%i1) f1(t):=6$      f2(t):=3$
      f(t):=if t<=2 then f1(t) else f2(t);
(%o3) f(t):=if t<=2 then f1(t) else f2(t)
(%i4) integrate(f1(t), t, 0, x);
      integrate(f2(t), t, 2, x);
(%o4) 6 x
      (%o5) 3 (x-2)

(%i6) f1(x):=6$      f2(x):=3$
      f(x):=if x<=2 then f1(t) else f2(t)$
(%i9) F1(x):=6*x$      F2(x):=F1(2)+3*(x-2)$
      F(x):=if x<=2 then F1(x) else F2(x);
(%o11) F(x):=if x<=2 then F1(x) else F2(x)
(%i12) wxplot2d([f(x), F(x)], [x, 0, 4], [y, 0, 25], [legend, "f(x)", "F(x)"], [style, [lines, 2]],
      [plot_format, gnuplot], [gnuplot_preamble, "set size ratio 1; set grid"], [nticks, 300])$
```

(%t12)



Aplicacions de la funció integral.

La definició de la funció integral permet calcular derivades de funcions definides mitjançant integrals, és a dir, funcions de la forma:

$$G(x) = \int_a^{u(x)} f(t)dt ; \quad H(x) = \int_{v(x)}^{u(x)} f(t)dt$$

En efecte, aplicant la regla de la cadena és immediat establir que la derivada d'aquestes funcions és:

$$DG(x) = D\left(\int_a^{u(x)} f(t)dt\right) = f(u(x)) Du(x)$$

$$DH(x) = D\left(\int_{v(x)}^{u(x)} f(t)dt\right) = f(u(x)) Du(x) - f(v(x)) Dv(x)$$

Vegem com es fan els càlculs amb wxMaxima.

Exemple 9.3.3. Es tracta de calcular la derivada de la funció definida per:

$$G(x) = \int_2^{e^x} \frac{\log(1+t)}{2t^2} dt$$

Aleshores:

```
(%i1) f(t):=(log(1+t))/(2*t^2);
      F(x):=integrate(f(t), t, 2, %e^x);
(%o1) f(t):=

$$\frac{\log(1+t)}{2t^2}$$

(%o2) F(x):=

$$\int_2^{e^x} f(t)dt$$

(%i3) assume(%e^x - 2 > 0)$
      DF(x)=diff(F(x), x);
(%o4) DF(x)=

$$\frac{{e^{-x}} (({e^x} + 1) \log({e^x} + 1) - x {e^x}) - {e^{-x}} ({e^x} \log({e^x} + 1) - x {e^x})}{2}$$

(%i5) ratsimp(%);
(%o5) DF(x)=

$$\frac{{e^{-x}} \log({e^x} + 1)}{2}$$

```

Exemple 9.3.4. Es tracta de calcular la derivada de la funció definida per:

$$H(x) = \int_{x+1}^{3x^2+2x-1} (1 + e^t) dt$$

Aleshores:

```
(%i1) F3(x):=integrate(f(t), t, u(x), v(x));
```

```
(%o1) F3(x):=  $\int_{u(x)}^{v(x)} f(t)dt$ 
```

```
(%i2) f(t):=exp(t)+1$    u(x):=x+1$    v(x):=3*x^2+2*x-1$  
diff(F3(x),x);
```

```
(%o5) %e-1 ((6 x + 2) %e3 x2 + 2 x + 6 %ex + 2 %e) - %ex + 1 - 1
```

```
(%i6) ratsimp(%);
```

```
(%o6) %e-1 ((6 x + 2) %e3 x2 + 2 x - %ex + 2 + 6 %ex + %e)
```

09-4.- Càlcul aproximat d'integrals. Integració numèrica.

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_09-4.wxm**.

Si $f : [a, b] \subset \mathbb{R} \rightarrow \mathbb{R}$ és integrable en aquest interval, el Teorema fonamental del Càlcul permet calcular la integral de la funció en $[a, b] \subset \mathbb{R}$. No obstant això, hi ha funcions per a les quals el càlcul de la seva integral indefinida és molt difícil o fins i tot no és expressable mitjançant funcions elementals, com és per exemple el cas de la funció $\exp(-x^2)$ que és contínua en qualsevol interval $[a, b] \subset \mathbb{R}$ i, per tant, admet una primitiva. Per poder resoldre el càlcul de la integral en aquests casos, es poden aplicar mètodes aproximats, com els que s'estudien en aquest apartat.

Si $f : [a, b] \subset \mathbb{R} \rightarrow \mathbb{R}$ és una funció integrable en $[a, b] \subset \mathbb{R}$, s'anomena partició de l'interval un conjunt de punts $P = \{x_0, x_1, \dots, x_n\}$ de l'interval $[a, b]$ tal que $h = x_{k+1} - x_k, k = 0, 1, \dots, n-1$. Naturalment es compleix $h = \frac{b-a}{n}$.

9.4.1. El mètode dels trapezis.

En cada un dels intervals $I_k = [x_k, x_{k+1}]$, $k = 0, 1, \dots, n-1$ associats a la partició es considera la funció g_k definida per la recta que passa pels punts $(x_k, f(x_k))$ i $(x_{k+1}, f(x_{k+1}))$; fet això, la integral de la funció donada en l'interval I_k es pot aproximar per la integral d'aquesta funció lineal en I_k . Finalment, la integral de la funció en $[a, b] \subset \mathbb{R}$ es pot aproximar per la suma de les integrals de g_k en I_k :

$$\int_a^b f(x) dx \approx \sum_{k=0}^{n-1} \int_{x_k}^{x_{k+1}} g_k(x) dx = h \left(\frac{f(a) + f(b)}{2} + \sum_{k=1}^{n-1} f(a + kh) \right)$$

expressió que és coneix com *fórmula dels trapezis*.

Exemple 9.4.1. Es vol calcular aproximadament per la fórmula dels trapezis la integral de la funció $f(x) = 1 + \frac{\cos(x)}{100 - x^6}$ en l'interval $[0, 2]$.


Farem els càlculs afinant successivament la partició de l'interval veient en primer lloc que el programa no ens dóna cap resultat si demanem calcular la integral:

```
(%i1) f(x):=1 + cos(x)/(100-x^6)$
      'integrate(f(x), x, 0, 2)=integrate(f(x), x, 0, 2);
```

$$(\%o2) \int_0^2 \frac{\cos(x)}{100-x^6} + 1 dx = \int_0^2 \frac{\cos(x)}{100-x^6} + 1 dx$$

Farem una representació gràfica de la funció en l'interval considerat:

```
(%i3) wxplot2d([f(x)], [x, 0, 2], [y, 0.96, 1.04], [ylabel, ""], [legend, "f(x)"], [style, [lines, 2]],
      [plot_format, gnuplot], [gnuplot_preamble, "set size ratio 1; set grid"] )$
```



I ara ja podem començar les iteracions:

```
(%i4) h=h:0.2$      n1=2/h;
```

```
(%o5) n1 = 10.0
```

```
(%i6) IA_trapezis_1=T1:h*((f(0) + f(2))/2 + sum(f(h*n), n, 1, 9)), numer;
```

```
(%o6) IA_trapezis_1 = 2.008142250402611
```

```
(%i7) kill(h)$      h=h:0.1$      n2=2/h;
```

```
(%o9) n2 = 20.0
```

```
(%i10) IA_trapezis_2=T2:h*((f(0) + f(2))/2 + sum(f(h*n), n, 1, 19)), numer;
```

```
(%o10) IA_trapezis_2 = 2.008332190715854
```

```
(%i11) kill(h)$      h=h:0.05$      n3=2/h;
```

```
(%o13) n3 = 40.0
```

```
(%i14) IA_trapezis_3=T3:h*((f(0) + f(2))/2 + sum(f(h*n), n, 1, 39)), numer;
```

```
(%o14) IA_trapezis_3 = 2.008384205938344
```

```
(%i15) kill(h)$      h=h:0.01$      n4=2/h;
(%o17) n4 = 200.0
(%i18) IA_trapezis_4=T4:h*((f(0) + f(2))/2 + sum(f(h*n), n, 1, 199)), numer;
(%o18) IA_trapezis_4 = 2.008401415164511

(%i19) kill(h)$      h=h:0.005$      n5=2/h;
(%o21) n5 = 400.0
(%i22) IA_trapezis_5=T5:h*((f(0) + f(2))/2 + sum(f(h*n), n, 1, 399)), numer;
(%o22) IA_trapezis_5 = 2.008401958078575
```

9.4.2. El mètode dels rectangles.

En cada un dels intervals $I_k = [x_k, x_{k+1}]$, $k = 0, 1, \dots, n-1$ associats a la partició es considera la funció constant g_k definida pel valor de la funció en el punt mig d'aquest interval, és a dir:

$$g_k(x_k^*) = f\left(\frac{x_k + x_{k+1}}{2}\right), \quad k = 0, 1, 2, \dots, n-1$$

Fet això, la integral de la funció donada en l'interval I_k es pot aproximar per la integral d'aquesta funció constant en I_k . Finalment, la integral de la funció en $[a, b] \subset \mathbb{R}$ es pot aproximar per la suma de les integrals de g_k en I_k :

$$\int_a^b f(x)dx \approx \sum_{k=0}^{n-1} \int_{x_k}^{x_{k+1}} g_k(x)dx = hf(x_k^*)$$

expressió que és coneix com *fórmula dels rectangles*.

Exemple 9.4.2. Es vol calcular aproximadament per la fórmula dels rectangles la integral de la funció $f(x) = 1 + \frac{\cos(x)}{100 - x^6}$ en l'interval $[0, 2]$.

Ja sabem que el programa no calcula aquesta integral i tenim la representació gràfica feta en l'Exemple anterior. Ja podem començar directament les iteracions:

```
(%i23) kill(h)$      h=h:0.2$      n1=2/h;
(%o25) n1 = 10.0
(%i26) IA_rectangles_1=R1:h*(sum(f(h*(n + n+1)/2), n, 0, 9)), numer;
(%o26) IA_rectangles_1 = 2.008522131029097
```

```

(%i27) kill(h)$      h=h:0.1$      n2=2/h;
(%o29) n2 = 20.0
(%i30) IA_rectangles_2=R2:h*(sum(f(h*(n + n+1)/2), n, 0, 19)), numer;
(%o30) IA_rectangles_2 = 2.008436221160835

(%i31) kill(h)$      h=h:0.05$      n3=2/h;
(%o33) n3 = 40.0
(%i34) IA_rectangles_3=R3:h*(sum(f(h*(n + n+1)/2), n, 0, 39)), numer;
(%o34) IA_rectangles_3 = 2.008411041591628

(%i35) kill(h)$      h=h:0.01$      n4=2/h;
(%o37) n4 = 200.0
(%i38) IA_rectangles_4=R4:h*(sum(f(h*(n + n+1)/2), n, 0, 199)), numer;
(%o38) IA_rectangles_4 = 2.00840250099264

(%i35) kill(h)$      h=h:0.01$      n4=2/h;
(%o37) n4 = 200.0
(%i38) IA_rectangles_4=R4:h*(sum(f(h*(n + n+1)/2), n, 0, 199)), numer;
(%o38) IA_rectangles_4 = 2.00840250099264

```

9.4.3. El mètode de Simpson.

Aquest mètode es basa en l'aproximació de la funció per un polinomi de segon grau. Per tal de poder dur a terme aquesta aproximació es requereix que el nombre d'interval·ls sigui parell i, per tant, que el nombre de punts de la partició sigui senar. Per tant, indicarem $n = 2p$, $h = \frac{b-a}{2p}$ i els punts de la partició seran $P = \{x_0, x_1, \dots, x_{2p}\}$.

L'expressió que resulta és:

$$\int_a^b f(x)dx \approx \frac{h}{3} \sum_{k=0}^{p-1} (f(x_{2k}) + 4f(x_{2k+1}) + f(x_{2k+2}))$$

expressió que és coneix com *fórmula de Simpson*.

Exemple 9.4.3. Es vol calcular aproximadament per la fórmula dels rectangles la integral de la funció $f(x) = 1 + \frac{\cos(x)}{100 - x^6}$ en l'interval $[0, 2]$.

Ja sabem que el programa no calcula aquesta integral i tenim la representació gràfica feta abans. Ja podem començar directament les iteracions:

```

(%i43) kill(h)$      h=h:0.2$      n1=n1:2/h;  p1=n1/2;
(%o45) n1 = 10.0
(%o46) p1 = 5.0
(%i47) IA_Simpson_1=
      S1:(h/3)*(sum(f(h*2*n) + 4*f(h*(2*n+1)) + f(h*(2*n+2)), n, 0, 4)), numer;
(%o47) IA_Simpson_1 = 2.008349561013199

(%i48) kill(h)$      h=h:0.1$      n2=n2:2/h;  p2=n2/2;
(%o50) n2 = 20.0
(%o51) p2 = 10.0
(%i52) IA_Simpson_2=
      S2:(h/3)*(sum(f(h*2*n) + 4*f(h*(2*n+1)) + f(h*(2*n+2)), n, 0, 9)), numer;
(%o52) IA_Simpson_2 = 2.008395504153601

(%i53) kill(h)$      h=h:0.02$      n3=n3:2/h;  p3=n3/2;
(%o55) n3 = 100.0
(%o56) p3 = 50.0
(%i57) IA_Simpson_3=
      S3:(h/3)*(sum(f(h*2*n) + 4*f(h*(2*n+1)) + f(h*(2*n+2)), n, 0, 49)), numer;
(%o57) IA_Simpson_3 = 2.008402121260181

(%i58) kill(h)$      h=h:0.01$      n4=n4:2/h;  p4=n4/2;
(%o60) n4 = 200.0
(%o61) p4 = 100.0
(%i62) IA_Simpson_4=
      S4:(h/3)*(sum(f(h*2*n) + 4*f(h*(2*n+1)) + f(h*(2*n+2)), n, 0, 99)), numer;
(%o62) IA_Simpson_4 = 2.008402137973978

```

9.4.4. Mètodes implementats en wxMaxima.

El programa wxMaxima té implementat l'anomenat mètode de Romberg, que és un mètode iteratiu basat en la fórmula dels trapezis (vegeu referències). La sintaxi és molt senzilla:

```

(%i63) 'integrate(f(x), x, 0, 2)=romberg(f(x), x, 0, 2);
(%o63)  $\int_0^2 \frac{\cos(x)}{100-x^6} + 1 \, dx = 2.008395056340455$ 

```

09-5.- Integrals impròpies.

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_09-5.wxm**.

Les integrals impròpies són una generalització de la integral d'una funció en un interval, en els casos següents:

- 1) L'interval no és acotat i la funció és acotada en aquest interval (integrals impròpies de primera espècie).
- 2) L'interval és tancat i la funció no és acotada en aquest interval (integrals impròpies de segona espècie).

Poden donar-se conjuntament les dues situacions anteriors, aplicant-se aleshores una metodologia que contempla els dos casos.

9.5.1 Integrals impròpies de primera espècie.

Considerem una funció $f : [a, +\infty[\rightarrow \mathbb{R}$ que compleix:

- La funció és acotada en el seu domini;
- La funció és integrable en tot interval $[a, x] \subset [a, +\infty[$ qualsevol que sigui $x > a$.

En aquestes condicions es defineix la integral (impròpia) de la funció en l'interval $[a, +\infty[$ mitjançant el límit a l'infinit de la seva funció integral, és a dir:

$$\int_a^{+\infty} f(t)dt = \lim_{x \rightarrow +\infty} \int_a^x f(t)dt$$

Si el límit existeix i és finit es diu que la integral (impròpia) és convergent. En cas contrari, es diu que és divergent.

Es defineixen de forma anàloga:

$$\begin{aligned} \int_{-\infty}^b f(t)dt &= \lim_{x \rightarrow -\infty} \int_x^b f(t)dt \\ \int_{-\infty}^{+\infty} f(t)dt &= \lim_{x \rightarrow +\infty} \int_c^x f(t)dt + \lim_{x \rightarrow -\infty} \int_x^c f(t)dt \end{aligned}$$

Exemple 9.5.1. Es considera la funció $f(x) = \exp(-ax)$, $a > 0$ definida en l'interval $[0, +\infty[$. Es tracta d'il·lustrar el mètode d'estudi de la convergència d'aquesta integral impròpia i, si és convergent, calcular-la.

En primer lloc definim la funció:

```
(%i1) f1(t):=exp(-a*t);
(%o1) f1(t):=exp((-a)t)
```

Ara calculem la funció integral:

```
(%i2) assume(a>0)$
'integrate(f1(t), t, 0, x)=integrate(f1(t), t, 0, x);
(%o3)  $\int_0^x e^{-a t} dt = \frac{1}{a} - \frac{e^{-a x}}{a}$ 
(%i4) F1(x):=1/a-%e^(-a*x)/a;
(%o4) F1(x):= $\frac{1}{a} - \frac{e^{(-a) x}}{a}$ 
```

Ara calculem el límit de la funció integral:

```
(%i5) 'limit(F1(x), x, inf)=limit(F1(x), x, inf);
(%o5)  $\lim_{x \rightarrow \infty} \frac{1}{a} - \frac{e^{-a x}}{a} = \frac{1}{a}$ 
```

Atès que és finit, la integral impròpia és convergent i val $1/a$. La sintaxi de wxMaxima que permet els càlculs directament és:

```
(%i6) 'integrate(f1(t), t, 0, inf)=integrate(f1(t), t, 0, inf);
(%o6)  $\int_0^{\infty} e^{-a t} dt = \frac{1}{a}$ 
```

Exemple 9.5.2. Es considera la funció $f(x) = \frac{1}{x^a}$, $a \geq 1$ definida en l'interval $[1, +\infty[$.

Es tracta d'il·lustrar el mètode d'estudi de la convergència d'aquesta integral impròpia i, si és convergent, calcular-la.

En primer lloc definim la funció:

```
(%i1) f2(t):=1/t^a;
(%o1) f2(t):= $\frac{1}{t^a}$ 
```

Dividirem l'estudi en dos casos: $a > 1$ i $a = 1$. Ara calculem la funció integral en el primer cas:

```
(%i2) assume(a>1)$      assume(x>1)$
      'integrate(f2(t), t, 1, x)=integrate(f2(t), t, 1, x);

(%o4)  $\int_1^x \frac{1}{t^a} dt = \frac{1}{a-1} - \frac{x^{1-a}}{a-1}$ 

(%i5) F2(x):=1/(a-1)-x^(1-a)/(a-1);

(%o5) F2(x):= $\frac{1}{a-1} - \frac{x^{1-a}}{a-1}$ 
```

Ara calculem el límit de la funció integral:

```
(%i6) 'limit(F2(x), x, inf)=limit(F2(x), x, inf);

(%o6)  $\lim_{x \rightarrow \infty} \frac{1}{a-1} - \frac{x^{1-a}}{a-1} = \frac{1}{a-1}$ 
```

Atès que és finit, la integral impròpia és convergent i val $1/a-1$. La sintaxi de wxMaxima que permet els càlculs directament és:

```
(%i7) 'integrate(f2(t), t, 1, inf)=integrate(f2(t), t, 1, inf);

(%o7)  $\int_1^{\infty} \frac{1}{t^a} dt = \frac{1}{a-1}$ 
```

Vegem ara el cas $a = 1$:

```
(%i8) f21(t):=1/t$
      'integrate(f21(t), t, 1, inf)=integrate(f21(t), t, 1, inf);
defint: integral is divergent.
-- an error. To debug this try: debugmode(true);
```

En aquest cas la integral impròpia és divergent.

9.5.2 Integrals impròpies de segona espècie.

Considerem una funció $f :]a, b] \rightarrow \mathbb{R}$ tal que:

- és integrable en tot interval tancat $[a + \delta, b] \subset]a, b]$ qualsevol que sigui $0 < \delta < b - a$;
- es compleix $\lim_{x \rightarrow a^+} f(x) = +\infty$.

Aleshores, es defineix la integral impròpia de f en l'interval $[a, b]$ mitjançant:

$$\int_a^b f(t)dt = \lim_{\delta \rightarrow 0} \int_{a+\delta}^b f(t)dt$$

Si aquest límit existeix, la integral impròpia es diu que és *convergent*; en cas contrari, es diu que és *divergent*.

Es defineix de forma anàloga la integral impròpia de la funció f en l'interval $[a, b]$, en els casos:

- La funció és integrable en tot interval tancat $[a, b - \delta] \subset [a, b[$ qualsevol que sigui $0 < \delta < b - a$ i $\lim_{b^-} f = \infty$
- La funció és integrable en els intervals $[a, x_0 - \delta]$ i $[x_0 + \delta, b]$ qualsevol que sigui $0 < \delta < \max\{x_0 - a, b - x_0\}$ i $\lim_{x_0^-} f = \lim_{x_0^+} f = \infty$.

Exemple 9.5.3. Es tracta d'estudiar la convergència de la integral impròpia

$$\int_1^2 \frac{1}{\sqrt{x-1}} dx$$

En primer lloc definim la funció:

```
(%i1) f4(t):=1/(sqrt(t-1));
(%o1) f4(t):= $\frac{1}{\sqrt{t-1}}$ 
```

Ara calculem la funció integral:

```
(%i2) assume(x-2<0)$      assume(x-1>0)$
      'integrate(f4(t), t, x, 2)=integrate(f4(t), t, x, 2);
(%o4)  $\int_x^2 \frac{1}{\sqrt{t-1}} dt = 2 - 2\sqrt{x-1}$ 
(%i5) F4(x):=2-2*sqrt(x-1);
(%o5) F4(x):= $2 - 2\sqrt{x-1}$ 
```

I finalment calculem el límit lateral per la dreta a l'extrem inferior de l'interval:

```
(%i6) 'limit(F4(x), x, 1, plus)=limit(F4(x), x, 1, plus);
(%o6)  $\lim_{x \rightarrow 1^+} 2 - 2\sqrt{x-1} = 2$ 
```

Per tant, la integral impròpia és convergent. La sintaxi per calcular-la amb wxMaxima és:

```
(%i7) 'integrate(f4(t), t, 1, 2)=integrate(f4(t), t, 1, 2);
```

```
(%o7)  $\int_1^2 \frac{1}{\sqrt{t-1}} dt = 2$ 
```

Observació. Considerem una funció $f :]a, b] \rightarrow \mathbb{R}$ tal que f és integrable en tot interval tancat $[a + \delta, b] \subset]a, b]$ qualsevol que sigui $0 < \delta < b - a$ i es compleix $\lim_{x \rightarrow a+} f(x) = +\infty$. Si es considera el canvi de variable definit per:

$$x = g(t) = a + \frac{1}{t}; \quad Dg(t) = -\frac{1}{t^2}$$

aleshores es compleix:

$$\int_a^b f(x) dx = \int_{+\infty}^{\frac{1}{b-a}} f\left(a + \frac{1}{t}\right) \left(-\frac{1}{t^2}\right) dt = \int_{\frac{1}{b-a}}^{+\infty} f\left(a + \frac{1}{t}\right) \left(\frac{1}{t^2}\right) dt$$

és a dir, la integral impròpia de segona espècie s'ha transformat en una integral impròpia de primera espècie.